

# Re: login script for group

---

*Source:*

[http://www.tech-archive.net/Archive/Windows/microsoft.public.windows.server.active\\_directory/2007-09/msg00949](http://www.tech-archive.net/Archive/Windows/microsoft.public.windows.server.active_directory/2007-09/msg00949)

---

- *From:* "Richard Mueller [MVP]" <[rlmueller-nospam@xxxxxxxxxxxxxxxxxxxxxxxxx](mailto:rlmueller-nospam@xxxxxxxxxxxxxxxxxxxxxxxxx)>
  - *Date:* Thu, 20 Sep 2007 13:10:35 -0500
- 

Deena wrote:

I would like to write a script that maps a drive to a shared folder for users who belong to a group. I have created a shared folder named "geometry" and have created a group (also called "geometry") and assigned users to it. I am trying the following script which I have saved to the SYSVOL on the server/domain/scripts folder. It is also saved to the NETLOGON folder although I didn't put it there – it just appeared after I put it in the SYSVOL directory. I am getting an error message on the strGroups = LCase(Join(CurrentUser.MemberOf)) line. Something about the use of Join. I copied this code from the Microsoft Tech website and simply changed the names of the group and server. Any suggestions?

Sample of my script:

```
Const geometry = "cn=Geometry"

Set ADSysInfo = CreateObject("ADSystemInfo")
Set CurrentUser = GetObject("LDAP:// & ADSysInfo.UserName")
strGroups = LCase(Join(CurrentUser.MemberOf))

If InStr(strGroups, geometry) Then

  wshNetwork.MapNetworkDrive "g:", "\\EAGLE4W1\Geometry\"

End If
```

The Microsoft script is flawed. The Join method expects an array, so an error is raised if the memberOf collection does not have at least two Distinguished Names. If the user is a member of at least two groups (not counting the "primary" group of the user, which is never included), then CurrentUser.memberOf will be an array and all is fine. If the user is a member of one group, CurrentUser.memberOf will be a string and Join will

## Re: login script for group

raise an error. If the user is not a member of any groups (except the "primary"), then `CurrentUser.memberOf` will be Empty and again an error is raised.

Note also that searching for a group name is not very reliable. If the name of the group is geometry, any group with the string "geometry" anywhere in the Distinguished Name will return True (an integer greater than 0). Another group could have a similar name, have the same name but be in another OU, or be in an OU with the string.

I discuss this and better methods to check group membership in this link:

<http://www.rlmueller.net/MemberOf.htm>

The method I would suggest for you is to bind to the group object and use the `IsMember` method. For example:

=====

Option Explicit

```
Dim ADSysInfo, CurrentUser, wshNetwork
Dim strGroupDN, objGroup
```

```
Set ADSysInfo = CreateObject("ADSystemInfo")
Set CurrentUser = GetObject("LDAP:// & ADSysInfo.UserName")
Set wshNetwork = CreateObject("Wscript.Network")
```

```
' Specify Distinguished Name of group.
strGroupDN = "cn=Geometry,ou=Sales,ou=West,dc=MyDomain,dc=com"
```

```
' Bind to the group object.
Set objGroup = GetObject("LDAP:// & strGroupDN")
```

```
' Check direct group membership.
If (objGroup.IsMember(CurrentUser.AdsPath) = True) Then
wshNetwork.MapNetworkDrive "g:", "\\EAGLE4W1\Geometry\"
End If
```

=====

Yes, this involves using the group Distinguished Name, but is really the only way to uniquely indentify the group. I also recommend using Option Explicit so you must declare all variables.

==

Richard Mueller  
Microsoft MVP Scripting and ADSI  
Hilltop Lab – <http://www.rlmueller.net>

==

^